

§ 3 Résolution d'équations avec Mathematica

Mathematica dispose de plusieurs méthodes pour résoudre des équations. Dans cette section, nous allons nous intéresser aux méthodes suivantes : **Solve**, **Reduce**, **NSolve** et **FindRoot**.

Avant d'aborder les spécificités de ces différentes fonctions, nous allons dans le préliminaire qui suit faire un bref rappel sur les symboles d'affectation et d'égalité puis, afin de comprendre quelle méthode est la plus appropriée à la résolution de quelle équation, nous allons proposer une classification des équations. Nous allons ensuite traiter les méthodes permettant d'obtenir des solutions exactes puis celles permettant d'obtenir des solutions approximatives.

■ § 3.1 Préliminaire

■ Symboles d'affectation et d'égalité

En *Mathematica*, le symbole “=” désigne une affectation (ou assignation) qui consiste à attribuer une valeur à une variable. Par exemple

```
x = 5
5
```

Par contre, “==” désigne le symbole d'égalité dont l'évaluation dans une expression retourne la valeur est vrai (**True**) ou faux (**False**) :

```
x == 1
False

x == 5
True
```

C'est le symbole d'égalité qui permet de poser une équation. L'équation comporte habituellement une ou plusieurs inconnue(s), c'est-à-dire une ou plusieurs variable(s) non affectée(s) :

```
Clear[x];
eqn =  $\frac{1}{2}x + \frac{2}{3} == 5$ 
 $\frac{2}{3} + \frac{x}{2} == 5$ 
```

Comme 0 n'est pas une solution de cette équation, si nous remplaçons **x** par 0 dans **eqn** *Mathematica* retourne faux :

```
eqn /. {x -> 0}
False
```

■ Equations algébriques et transcendantes

Une équation algébrique est une équation qui peut être mise sous la forme $p(x) = 0$, où p est un polynôme. Par exemple, l'équation $\frac{x+1}{x+7} = 2$ est algébrique car elle est équivalente à $(x+1) - 2(x+7) = 0$ sous la condition $x \neq -7$. Une équation qui n'est pas algébrique est une équation transcendante. Par exemple, $\sin(\alpha) = \cos(\alpha)$ est transcendante.

Afin d'illustrer le fonctionnement des méthodes de résolution proposées par *Mathematica* nous allons travailler avec les équations suivantes :

```
Clear[x, a, b]
(* Equations algébriques *)
eqnA1 = x^5 + x + 1 == 0;
eqnA2 = x^5 + x^2 + 1 == 0;
eqnA3 = a x + b == 0;
(* Equations transcendantes *)
eqnT1 = Cos[x] == 1 / 2;
eqnT2 = Cos[x] == Log[x];
```

■ § 3.2 Recherche des solutions exactes

■ Méthode `Solve`

La méthode `Solve` est spécialement conçue pour la résolution d'équations algébriques mais fonctionne aussi pour certaines équations transcendantes. Nous allons voir comment *Mathematica* traite nos cinq exemples.

```
Solve[eqnA1, x]
```

$$\left\{ \left\{ x \rightarrow -(-1)^{1/3} \right\}, \left\{ x \rightarrow (-1)^{2/3} \right\}, \left\{ x \rightarrow \frac{1}{3} \left(1 - \left(\frac{2}{25 - 3\sqrt{69}} \right)^{1/3} - \left(\frac{1}{2} (25 - 3\sqrt{69}) \right)^{1/3} \right) \right\}, \right.$$

$$\left. \left\{ x \rightarrow \frac{1}{3} + \frac{1}{6} (1 + i\sqrt{3}) \left(\frac{1}{2} (25 - 3\sqrt{69}) \right)^{1/3} + \frac{1 - i\sqrt{3}}{3 \times 2^{2/3} (25 - 3\sqrt{69})^{1/3}} \right\}, \right.$$

$$\left. \left\{ x \rightarrow \frac{1}{3} + \frac{1}{6} (1 - i\sqrt{3}) \left(\frac{1}{2} (25 - 3\sqrt{69}) \right)^{1/3} + \frac{1 + i\sqrt{3}}{3 \times 2^{2/3} (25 - 3\sqrt{69})^{1/3}} \right\} \right\}$$

Nous pouvons préciser que nous ne nous intéressons qu'aux solutions réelles :

```
Solve[eqnA1, x, Reals]
{{x -> Root[1 - #1^2 + #1^3 &, 1]}}
```

Mathematica nous dit que x est une racine de l'équation $1 - x^2 + x^3 = 0$. Afin d'obtenir cette solution sous forme de radicaux, nous utilisons la fonction `ToRadicals` :

```
ToRadicals[%]
{{x -> 1/3 (1 - (2/(25 - 3*sqrt(69)))^(1/3) - (1/2 (25 - 3*sqrt(69)))^(1/3))}}
```

```
Solve[eqnA2]
{{x -> Root[1 + #1^2 + #1^5 &, 1]}, {x -> Root[1 + #1^2 + #1^5 &, 2]},
 {x -> Root[1 + #1^2 + #1^5 &, 3]}, {x -> Root[1 + #1^2 + #1^5 &, 4]}, {x -> Root[1 + #1^2 + #1^5 &, 5]}}
```

```
ToRadicals[%]
{{x -> Root[1 + #1^2 + #1^5 &, 1]}, {x -> Root[1 + #1^2 + #1^5 &, 2]},
 {x -> Root[1 + #1^2 + #1^5 &, 3]}, {x -> Root[1 + #1^2 + #1^5 &, 4]}, {x -> Root[1 + #1^2 + #1^5 &, 5]}}
```

Mathematica n'arrive pas à écrire les solutions de cette équation sous forme de radicaux. Ce n'est pas étonnant car nous avons une équation de degré 5 et il n'existe en effet pas de méthode générale pour résoudre les équation de degré 5.

```
Solve[eqnA3, x]
```

$$\left\{ \left\{ x \rightarrow -\frac{b}{a} \right\} \right\}$$

Mathematica donne la solution générale de l'équation mais ne traite pas le cas $a=0$.

```
Solve[eqnT1, x]
```

$$\left\{ \left\{ x \rightarrow \text{ConditionalExpression}\left[-\frac{\pi}{3} + 2\pi C[1], C[1] \in \text{Integers}\right] \right\}, \right. \\ \left. \left\{ x \rightarrow \text{ConditionalExpression}\left[\frac{\pi}{3} + 2\pi C[1], C[1] \in \text{Integers}\right] \right\} \right\}$$

Bien que l'équation soit transcendante, **Solve** donne toutes les solutions mais ceci est un cas particulièrement simple. Généralement **Solve** ne résout pas les équations transcendentes :

```
Solve[eqnT2, x]
```

Solve::nsmet: This system cannot be solved with the methods available to Solve. >>

```
Solve[Cos[x] == Log[x], x]
```

■ Méthode **Reduce**

La méthode **Reduce** réduit une équation à des équations équivalentes.

```
Reduce[eqnA1, x]
```

$$x == -(-1)^{1/3} \mid \mid x == (-1)^{2/3} \mid \mid x == \text{Root}[1 - \#1^2 + \#1^3 \&, 1] \mid \mid \\ x == \text{Root}[1 - \#1^2 + \#1^3 \&, 2] \mid \mid x == \text{Root}[1 - \#1^2 + \#1^3 \&, 3]$$

Là aussi, nous pouvons préciser que nous ne nous intéressons qu'aux solutions réelles :

```
Reduce[eqnA1, x, Reals]
```

$$x == \text{Root}[1 - \#1^2 + \#1^3 \&, 1]$$

Nous pouvons aussi utiliser la fonction **ToRadicals** :

```
ToRadicals[%]
```

$$x == \frac{1}{3} \left(1 - \left(\frac{2}{25 - 3\sqrt{69}} \right)^{1/3} - \left(\frac{1}{2} (25 - 3\sqrt{69}) \right)^{1/3} \right)$$

Nous pouvons transformer cette équation en une règle de remplacement avec la fonction **ToRules** (ceci est utile pour récupérer un résultat sans faire de copier-coller) :

```
ToRules[%]
```

$$\left\{ x \rightarrow \frac{1}{3} \left(1 - \left(\frac{2}{25 - 3\sqrt{69}} \right)^{1/3} - \left(\frac{1}{2} (25 - 3\sqrt{69}) \right)^{1/3} \right) \right\}$$

```
Reduce[eqnA2]
```

$$x == \text{Root}[1 + \#1^2 + \#1^5 \&, 1] \mid \mid x == \text{Root}[1 + \#1^2 + \#1^5 \&, 2] \mid \mid \\ x == \text{Root}[1 + \#1^2 + \#1^5 \&, 3] \mid \mid x == \text{Root}[1 + \#1^2 + \#1^5 \&, 4] \mid \mid x == \text{Root}[1 + \#1^2 + \#1^5 \&, 5]$$

```
ToRadicals[%]
```

$$x == \text{Root}[1 + \#1^2 + \#1^5 \&, 1] \mid \mid x == \text{Root}[1 + \#1^2 + \#1^5 \&, 2] \mid \mid \\ x == \text{Root}[1 + \#1^2 + \#1^5 \&, 3] \mid \mid x == \text{Root}[1 + \#1^2 + \#1^5 \&, 4] \mid \mid x == \text{Root}[1 + \#1^2 + \#1^5 \&, 5]$$

Comme avec **Solve**, *Mathematica* n'arrive pas à écrire les solutions de cette équation sous forme de radicaux.

```
Reduce[eqnA3, x]
```

$$(b == 0 \ \&\& \ a == 0) \ || \ (a \neq 0 \ \&\& \ x == -\frac{b}{a})$$

Pour comprendre la signification de ce résultat, rappelons que **&&** signifie “et” et que **||** signifie “ou” (ces symboles peuvent être remplacés par \wedge et \vee). La réponse fournie par *Mathematica* signifie que nous avons trois cas possibles :

1^{er} cas: si $a = 0$ et $b = 0$, alors tout x est solution;

2^e cas: si $a \neq 0$, alors $x = -\frac{b}{a}$;

3^e cas $a = 0$ et $b \neq 0$ (qui n'apparaît pas formellement), l'équation n'a pas de solution.

```
Reduce[eqnT1, x]
```

$$C[1] \in \text{Integers} \ \&\& \ (x == -\frac{\pi}{3} + 2\pi C[1] \ || \ x == \frac{\pi}{3} + 2\pi C[1])$$

Bien que l'équation soit transcendante, **Reduce** donne toutes les solutions mais ceci est un cas particulièrement simple. Généralement **Reduce** ne résout pas les équations transcendentes :

```
Reduce[eqnT2, x]
```

```
Reduce::nsmet: This system cannot be solved with the methods available to Reduce. >>
```

```
Reduce[Cos[x] == Log[x], x]
```

■ Comparaison de **Solve** et **Reduce**

Nous constatons que les deux méthodes présentent les mêmes performances pour des équations sans paramètre. **Solve** a l'avantage de pouvoir facilement réutiliser les résultats car ils sont fournis sous la forme de règles de remplacement. Par contre **Reduce** est plus performante que **Solve** si l'équation à résoudre possède des paramètres.

■ § 3.3 Approximation des solutions

Dans cette section nous allons approximer les solutions des équations que nous n'avons pu résoudre exactement dans la section précédente.

■ Méthode **NSolve**

La méthode **NSolve** est spécialisée dans la résolution des équations algébriques (elle est équivalente à **Solve[eqn, x] / N**). Elle livrera une approximation de toutes les solutions.

```
NSolve[eqnA2, x]
```

$$\{\{x \rightarrow -1.19386\}, \{x \rightarrow -0.15459 - 0.828074 i\}, \{x \rightarrow -0.15459 + 0.828074 i\}, \\ \{x \rightarrow 0.751519 - 0.784616 i\}, \{x \rightarrow 0.751519 + 0.784616 i\}\}$$

Nous pouvons préciser que nous ne nous intéressons qu'aux solutions réelles :

```
NSolve[eqnA2, x, Reals]
```

$$\{\{x \rightarrow -1.19386\}\}$$

```
NSolve[eqnT2, x]
```

NSolve::nsmet: This system cannot be solved with the methods available to NSolve. >>


```
NSolve[Cos[x] == Log[x], x]
```

Sans surprise, **NSolve** ne résout pas le deuxième exemple d'équation transcendante.

■ Méthode **FindRoot**[eqn, {x, x0}]

La méthode **FindRoot**[eqn, {x, x0}] est une variation de la méthode de Newton. Pour que *Mathematica* retourne une approximation d'une solution de l'équation **eqn** dont l'inconnue est **x**, nous devons donner une approximation de départ **x0**. Appliquons cette méthode au deuxième exemple d'équation transcendante. Pour cela, nous devons trouver une approximation de départ, approximation que nous allons obtenir graphiquement :

```
Plot[{Cos[x], Log[x]}, {x, 0, 10}]
```



Nous pouvons par exemple prendre 1 comme première approximation :

```
FindRoot[eqnT2, {x, 1}]
```

```
{x -> 1.30296}
```

Vérifions notre solution :

```
{Cos[x], Log[x]} /. %
```

```
{0.264642, 0.264642}
```

Il est important d'avoir une approximation de notre solution qui soit suffisamment proche de la solution, sinon *Mathematica* n'arrive pas à trouver la solution :

```
FindRoot[eqnT2, {x, 4}]
```

FindRoot::lstol:

The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. >>

```
{x -> 6.11902}
```

```
{Cos[x], Log[x]} /. %
```

```
{0.986556, 1.8114}
```

■ Méthode `FindRoot[eqn, {x, x0, x1}]`

Pour certaines équations coriaces, il peut être utile de donner deux approximations x_0 et x_1 de la solution. C'est ce que fait la méthode `FindRoot[eqn, {x, x0, x1}]` qui est une variation de la méthode de Regula Falsi. Appliquons cette méthode au deuxième exemple d'équation transcendante.

```
FindRoot[eqnT2, {x, 4, 5}]
```

```
{x -> 1.30296}
```

Les deux approximations de départ sont mauvaises, mais *Mathematica* arrive tout de même à trouver la solution.

■ § 3.5 Exercices

■ Exercice 3 - P 1

D'un cône, on donne sa hauteur h et son aire totale A .

Calculez son rayon r en résolvant l'équation du problème 1 - P 1 avec *Mathematica*.

Indications

Il s'agit d'un calcul littéral.

Complétez l'équation par une condition afin de n'obtenir qu'une seule réponse.

■ Exercice 3 - P 2

D'un triangle rectangle, on connaît son aire A et la projection p d'une cathète sur l'hypoténuse.

Calculez les longueurs des trois côtés a , b , c en résolvant les équations du problème 1 - P 2 avec *Mathematica*.

Données numériques : $A = 975$, $p = 20$.

■ Exercice 3 - P 3

Une sphère homogène de rayon $r = 0.3 \text{ m}$, de masse volumique $\rho_1 = 900 \frac{\text{kg}}{\text{m}^3}$ flotte sur un liquide de masse volumique $\rho_0 = 1000 \frac{\text{kg}}{\text{m}^3}$.

Calculez la hauteur h de la partie émergente en résolvant l'équation du problème 1 - P 3 avec *Mathematica*.

■ Exercice 3 - P 4

Une cuve cylindrique est posée horizontalement sur le sol.

Donnée numérique $r = 50$.

Graduez une jauge, c'est-à-dire dessinez une bande de papier portant les graduations 10 %, 20 %, ... comme l'indique la figure ci-dessous.

— 10%

— 20%

— 30%

— 40%

— 50%

— 60%

— 70%

— 80%

— 90%

— 100%

— 0%

a) Première version: résoudre les équations l'une après l'autre, sans utiliser de listes

Indications

Il s'agit d'abord de résoudre 8 équations du problème 1 - P 4 et de former la liste des angles qui correspondent à chaque graduation:

$$\alpha_0 = 0;$$

α_1 est la solution de l'équation pour le taux de remplissage $t = 0.1$;

α_2 est la solution de l'équation pour le taux de remplissage $t = 0.2$;

α_3 est la solution de l'équation pour le taux de remplissage $t = 0.3$;

α_4 est la solution de l'équation pour le taux de remplissage $t = 0.4$;

$$\alpha_5 = \pi;$$

α_6 est la solution de l'équation pour le taux de remplissage $t = 0.6$;

α_7 est la solution de l'équation pour le taux de remplissage $t = 0.7$;

α_8 est la solution de l'équation pour le taux de remplissage $t = 0.8$;

α_9 est la solution de l'équation pour le taux de remplissage $t = 0.9$;

$$\alpha_{10} = 2\pi.$$

Il s'agit ensuite de former la liste des hauteurs qui correspondent à chaque angle

$$h_0 = 0;$$

h_1 est la hauteur qui correspond à l'angle α_1 ;

h_2 est la hauteur qui correspond à l'angle α_2 ;

h_3 est la hauteur qui correspond à l'angle α_3 ;

h_4 est la hauteur qui correspond à l'angle α_4 ;

$$h_5 = r;$$

h_6 est la hauteur qui correspond à l'angle α_6 ;

h_7 est la hauteur qui correspond à l'angle α_7 ;

h_8 est la hauteur qui correspond à l'angle α_8 ;

h_9 est la hauteur qui correspond à l'angle α_9 ;

$$h_{10} = 2r.$$

Il faut enfin dessiner une liste de traits horizontaux :

```
Show[Graphics[{
  Line[{{0,h0},{5, h0}}], Line[{{0,h1},{5, h1}}],
  Line[{{0,h2},{5, h2}}], Line[{{0,h3},{5, h3}}],
  Line[{{0,h4},{5, h4}}], Line[{{0,h5},{5, h5}}],
```

```

Line[{{0,h6},{5, h6}}], Line[{{0,h7},{5, h7}}],
Line[{{0,h8},{5, h8}}], Line[{{0,h9},{5, h9}}],
Line[{{0,h10},{5, h10}}], Text["0 %",{5,h0},{-1,0}],
Text["10 %",{5,h1},{-1,0}], Text["20 %",{5,h2},{-1,0}],
Text["30 %",{5,h3},{-1,0}], Text["40 %",{5,h4},{-1,0}],
Text["50 %",{5,h5},{-1,0}], Text["60 %",{5,h6},{-1,0}],
Text["70 %",{5,h7},{-1,0}], Text["80 %",{5,h8},{-1,0}],
Text["90 %",{5,h9},{-1,0}], Text["100 %",{5,h10},{-1,0}],
PlotRange->{{0,20},{-10,110}}, AspectRatio ->5]]];

```

b) Deuxième version : écrire un programme plus compact en utilisant Table[...]

■ Exercice 3 - P 5

Pour rembourser un emprunt de 8'200 francs, une banque demande de verser 6 annuités de 2'000 francs
 Calculez le taux d'intérêt i en résolvant l'équation du problème 1 - P 5 avec *Mathematica*.

Exercice de récapitulation

Un stère de bois (cube de 1 m de côté) est entassé contre une grande maison. A quelle distance x de la maison faut-il poser le pied d'une échelle de 10 m pour qu'elle s'appuie tant contre la façade que contre l'angle du tas de bois?

■ (a) Partie sans ordinateur

- (i) Déduire de l'énoncé que les valeurs possibles de x sont les zéros de la fonction

$$f(x) = (x-1)\sqrt{100-x^2} - x, x \in [1, 10].$$

- (ii) Déterminer la plus grande solution avec une erreur absolue inférieure 0.02 à l'aide de la méthode de la bisection.
 (iii) Déterminer la plus petite solution avec une précision de trois chiffres après la virgule à l'aide de la méthode regula falsi.

■ (b) Partie avec ordinateur

- (i) L'itération de la fonction $g(x) = f(x) + x$ converge-t-elle si on prend comme valeur initiale 1.5? Argumenter à l'aide d'un graphique.
 (ii) Déterminer avec la méthode pseudo Newton les solutions du problème avec une précision de 5 chiffres après la virgule.
 (iii) Déterminer avec une méthode appropriée de *Mathematica* les solutions du problème.

■ Réponses

- (a) (i) -
(ii) $x_{\max} \simeq (9.95 \pm 0.02) \text{ m}$
(iii) $x_{\min} \simeq 1.112 \text{ m}$
- (b) (i) La méthode diverge
(ii) La méthode donne $x_{\min} \simeq 1.11188 \text{ m}$ mais ne permet pas de trouver x_{\max} .
(iii) $x_{\min} = \frac{1}{2} (1 + \sqrt{101} - \sqrt{2(49 - \sqrt{101})}) \simeq 1.11188 \text{ m}$

$$x_{\max} = \frac{1}{2} + \frac{\sqrt{101}}{2} + \sqrt{\frac{1}{2}(49 - \sqrt{101})} \simeq 9.93799 \text{ m}$$